NCL
NCAR Command Language

DKRZ
DEUTSCHES
KLIMARECHENZENTRUM

NCL

SUPPLEMENT DOC

PLOT LAYOUT

High Quality Graphics
with
NCL 6.2.0

Karin Meier-Fleischer
DKRZ

Version: 1.0 2014/06/06



Contact: **Karin Meier-Fleischer**

Deutsches Klimarechenzentrum (DKRZ)
Bundesstrasse 45a
D-20146 Hamburg
Germany

Email: meier-fleischer(at)dkrz.de
http://www.dkrz.de/

The NCAR Command Language (Version 6.2.0) [Software]. (2014).
Boulder, Colorado: UCAR/NCAR/CISL/VETS. http://dx.doi.org/10.5065/D6WD3XH5

http://www.ncl.ucar.edu

Content

1 Introduction 4
2 Defaults 5
2.1 XY plot 5
2.2 XY plot using some resources 6
2.3 Contour plot defaults 8
2.4 Vector plot defaults 11
3 Titles 13
4 Add text 14
5 Function Codes 16
6 Tickmark settings 18
6.1 XY plot 18
6.2 Map plot 19
7 Legend Settings 20
8 Labelbar Settings 21
9 Insert a Logo 22
10 Change the size of the plot output file 23
11 Crop white space around the plot 24
12 Appendix A: Font Table 25
13 Appendix B: Text Alignment 26
14 Appendix C: Example scripts 27

1 Introduction

NCL provides a huge amount of graphic resources which allow the user to control the plot and its layout. Some resources are set by default from NCL like the left and right string on top of a contour plot or the labelbar at the bottom of a contour plot with fill mode on. Additional, NCL provides opportunities for the user to change or set more resources to get the best result in creating a plot which can be used for a presentation or a publication.

In this supplement documentation we will give you an introduction on how NCL resources can be used to control the layout of the plot output. Theoretical explainations followed by examples as easy as possible which demonstrate the capabilities like changing or setting the following:

- title font
- title font size
- font color
- change pre-defined text strings
- add text to plot
- positioning of text in a plot
- special characters e.g. ^{O}C or H_2O
- axis labeling
- tickmarks settings
- labelbar settings
- overlay a JPEG logo onto a plot
- crop white space around the plot
- change theb size of the plot output file



Precondition for working with this document is a basic knowledge of NCL graphics and its scripting language. How to write a script or how to plot your data are not made, only resource settings and some additional functions are described.

2 Defaults

The NCL plot functions starting with **gsn_csm** (gsm_csm_xy, gsn_csm_contour, gsn_csm_vector, etc**.)** are doing some default settings like adding axis labeling, a string on the left top of the plot and a string on the right top of the plot. Also for contour line plots a contouring reference box will be added at the right bottom of the plot and for a contour fill plot a labelbar will be added at the bottom of the plot. And drawing a vector plot a reference vector is plotted at the right bottom of the plot. NCL calculates the best position for the reference values, labelbars and axis labeling, but the user can change or delete this settings manually.

2.1 XY plot



2.2 XY plot using some resources

To change the x-axis and y-axis strings some resource settings has to be changed. The `@ti` in the resource name indicates that the 'Title' resources are used and the `XAxisString` tells you exactly what it is, a user defined string for the x-axis. To write a title string centered on top of the plot the `@tiMainString` can be set. The `@gsn` is the resource of the `'high-level graphical interface'` of NCL, the heart of it. The `gsnLeftString`, `gsnCenterString` and `gsnRightString` point to the reserved spaces for strings and in some cases they are set per default to the data attributes long_name and units (see contour and vector plots). To delete those pre-defined strings just define it to an empty string "" (res@gsnLeftString = "" avoid writing a left string at the top of the plot).



To resize the entire plot the **@vp** (viewport) resources can be used. With the **@vpXF** and **@vpYF** resources you can set the start drawing position and with the **@vpWidthF** and **@vpHeightF** the size of the plot space in the NDC plotting space can be set. But keep in mind that the space for the titles is not included and the titles and axis labels may be cut off.

Sometimes it is nice to have an auxiliarly grid of the NDC coordinates to find the right position for the titles, labels etc. This can be done by the following function call which must be called befor the plot function is called.

```
drawNDCGrid(wks)    ;-- draw the NDC grid
```



2.3 Contour plot defaults

Which defaults are used for a contour plot depends on the selected plotting function. The **gsn_csm_contour** function draws the x-axis in the same relation as the y-axis and it doesn't write the axis labels, but the top left and top right string are set per default as variable *long_name* and the variable's *units*. To set the x-axis and y-axis labels use the **@tiXAxisString** and **@tiYAxisString** resources (see before). If the units of the latitude and longitude variables are e.g. 'degrees_north' and 'degrees_east', NCL writes a single character next to the right of the axis labels which represents the geographic direction.



To create a contour plot using a correct relation between the x- and y-axis use the **gsn_csm_contour_map** function. It draws the x-axis and the y-axis in an appropriate relation for its projection and draws an underlying map in gray color, but it doesn't write the axis labels, too. If the units of the latitude and longitude variables are e.g. 'degrees_north' and 'degrees_east', NCL writes a single character next to the right of the axis labels which represents the geographic direction.

gsn_csm_contour_map



The default settings for a filled contour plot are very similar to the contour line plot settings as shown before except for the map which is just shown as black outlines of the continents and that a labelbar is created at the bottom of the plot. In the example below only one resource is set to get the filled contour plot:

```
res             =  True
res@cnFillOn    =  True
plot = gsn_csm_contour_map(wks,var,res)
```



2.4 Vector plot defaults

Also for vector plots the default settings depend on the selected plotting function. The **gsn_csm_vector** function draws the x-axis in the same relation as the y-axis and it doesn't write axis labels, but the top left and top right string are set per default as variable long_name and the variable's units. To set the x-axis and y-axis labels use the **@tiXAxisString** and **@tiYAxisString** resources (see before). If the units of the latitude and longitude variables are e.g. 'degrees_north' and 'degrees_east', NCL writes a single character next to the right of the axis values which represents the geographic direction.



To create a vector plot using a correct relation between the x- and y-axis use the **gsn_csm_vector_map** function. It draws the x-axis and the y-axis in an appropriate relation for its projection and draws an underlying map in gray color, but it doesn't write the axis labels, too. If the units of the latitude and longitude variables are e.g. 'degrees_north' and 'degrees_east', NCL writes a single character next to the right of the axis values which represents the geographic direction.

gsn_csm_vector_map



3 Titles

Now let us see how we can change, set, move and resize the main title, the axis titles, the strings above the plot and the labelbar title.

```
res@gsnLeftStringFontHeightF    =  0.008
res@gsnLeftStringParallelPosF   = -0.05
res@gsnLeftStringFontColor      = "red"
```

```
res@tiMainString = "This is a title line~C~~Z50~          with a continuing line"
res@tiMainFontHeightF =  0.05
res@tiMainFontColor   = "blue"
```

```
res@gsnCenterString   = "gsnCenterString"
res@gsnCenterStringFontHeightF =  0.013
res@gsnCenterStringFontColor   = "green"
```

```
res@gsnRightStringFontHeightF   =  0.015
res@gsnRightStringFontColor     = "darkorchid"
res@gsnRightStringParallelPosF =  1.01
```




```
res@tiYAxisString       = "tiYAxisString"
res@tiYAxisFontHeightF =  0.02
res@tiYAxisFontColor    = "navy"
res@tiYAxisSide         = "right"
res@tiYAxisAngleF       =  270
```

```
res@lbTitleOn           = True
res@lbTitleFontHeightF =  0.014
res@lbTitlePosition     = "Bottom"
res@lbTitleString     = "Temperature in Kelvin"
```

```
res@tiXAxisString       = "tiXAxisString"
res@tiXAxisFontHeightF =  0.02
res@tiXAxisFontColor    = "red"
```

4 Add text

Sometimes the settings decribed before does not fit your requirements and it is necessary to add more text to the plot. This can be done by the text function **gsn_add_text** which adds text to an existing plot (only in the plot area) or **gsn_text_ndc** which is able to write text at any place on the frame using the NDC coordinates.

```
txres@txFontColor              = "red"
txres@txFontHeightF            =  0.035
txres@txFontOpacityF           =  0.5
txres@txJust                   = "BottomLeft"
txres@txBackgroundFillColor    = -1

id = gsn_add_text(wks,plot,"gsn_add_text transparent",1,2,txres)
```

```
txres@txFontColor       = "blue"
txres@txFontHeightF     =  0.03
txres@txJust            = "CenterCenter"

id = gsn_add_text(wks, plot, "gsn_add_text", 14, 4,
txres)
```




```
txres@txFontColor              = "navy"
txres@txFontHeightF            =  0.02
txres@txAngleF                 = -45
txres@txBackgroundFillColor    = "yellow"

str = "gsn_add_text bg"

id = gsn_add_text(wks, plot, str, 8, 0, txres
```

```
value = 3.83927489235
str = "formatted value "+sprintf("%3.4f", value)+" in text"

txres@txFontColor       = "violet"
txres@txFontOpacityF    =  1.0

id = gsn_add_text(wks, plot, str, 1, -3.5, txres)
```

```
txres@txFontColor       = "green"
txres@txFontHeightF     =  0.02
txres@txAngleF          = 45

str = "gsn_add_text rotate 45~S~o~N~"

id = gsn_add_text(wks,plot,str,22,-1,txres)
```

```
ndcres@txFontHeightF      =  0.03
ndcres@txJust             = "BottomLeft"
gsn_text_ndc(wks, "Text line 1", 0.15, 0.95, ndcres)
ndcres@txFontHeightF      =  0.015
gsn_text_ndc(wks, "Text line 2", 0.15, 0.90, ndcres)
ndcres@txFontHeightF      =  0.01
gsn_text_ndc(wks, "Text line 3", 0.15, 0.87, ndcres)
```

```
ndcres@txFontHeightF     =  0.018
ndcres@txFontColor       = "blue"
ndcres@txAngleF          =  90
ndcres@txJust            = "CenterCenter"
gsn_text_ndc(wks, "y-axis title string", 0.09, 0.5, ndcres)

ndcres@txAngleF          = -90
gsn_text_ndc(wks, "y-axis title string", 0.85, 0.5, ndcres)

ndcres@txAngleF          =  0
gsn_text_ndc(wks, "x-axis title string", 0.5, 0.83, ndcres)
gsn_text_ndc(wks, "x-axis title string", 0.5, 0.12, ndcres)
```




```
ndcres@txFontHeightF     =  0.04
ndcres@txFontColor       = "orange"
ndcres@txJust            = "BottomLeft"

super = "35.5~S~o~N~"
sub   = "H~B~2~N~O"

gsn_text_ndc(wks, super, 0.23, 0.75, ndcres)

ndcres@txAngleF          = 45
gsn_text_ndc(wks, sub,   0.7, 0.25, ndcres)
```

```
ndcres@txFontColor       = "red"
ndcres@txFontHeightF     =  0.013
ndcres@txJust            = "BottomRight"
string = "gsn_text_ndc  -->  ~F35~c ~F21~~N~DKRZ 2014"
gsn_text_ndc(wks, string, 0.9, 0.02, ndcres)
```

```
Auml     = "A~H-15V6F35~H~FV-6H3~"
auml     = "a~H-13V2F35~H~FV-2H3~"
Ouml     = "O~H-16V6F35~H~FV-6H3~"
ouml     = "o~H-14V2F35~H~FV-2H3~"
Uuml     = "U~H-15V6F35~H~FV-6H3~"
uuml     = "u~H-13V2F35~H~FV-2H3~"
string = "Umlaute: "+Auml+" "+auml+"   "+Ouml+" "+ouml+"   "+Uuml+" "+uuml
ndcres@txFontColor       = "black"
ndcres@txFontHeightF     =  0.025
gsn_text_ndc(wks, string, 0.05, 0.06, ndcres)
```

5 Function Codes




```
  ndcres@gsnDraw                =  False              ;-- don't advance frame
  ndcres@gsnFrame               =  False              ;-- don't advance frame
  ndcres@txFontHeightF          =  0.03               ;-- decrease font size
  ndcres@txJust                 = "CenterCenter"      ;-- text justification
  ndcres@txFont                 = "helvetica-bold"    ;-- change font
  txid = gsn_create_text_ndc(wks,"Fonts and Function Codes", 0.5,0.96,ndcres)

  ndcres@txJust                 = "BottomLeft"        ;-- text justification
  ndcres@txFont                 = "helvetica"         ;-- change font
  line1 = "Font Height = 0.03"
  gsn_text_ndc(wks, line1, 0.09, 0.86, ndcres)
  line2 = "Font Height = 0.015"
  ndcres@txFontHeightF          =  0.015              ;-- decrease font size
  gsn_text_ndc(wks, line2, 0.09, 0.83, ndcres)
  line3 = "Font Height = 0.01"
  ndcres@txFontHeightF          =  0.01               ;-- decrease font size
  gsn_text_ndc(wks, line3, 0.09, 0.80, ndcres)

;-- superscripting
  ndcres@txFontHeightF          =  0.03               ;-- make font size larger
  ndcres@txFontColor            = "blue"              ;-- change to blue
  super = "Superscripting:  35.5~S~o~N~   x~S~2~N~   kgm~S~-2~N~s~S~-1~N~"
  gsn_text_ndc(wks, super, 0.09, 0.74, ndcres)

;-- subscripting
  sub   = "Subscripting:     H~B~2~N~O   CO~B~2~N~"
  gsn_text_ndc(wks, sub,   0.09, 0.69, ndcres)

;-- different font size in one string
  str_one = "Different ~Z70~font ~Z130~sizes~Z100~ in one ~Z60~string~Z100~"
  ndcres@txFontColor            = "seagreen"          ;-- change to seagreen
  gsn_text_ndc(wks, str_one, 0.09, 0.63, ndcres)

;-- Umlaute
  Auml  = "A~H-15V6F35~H~FV-6H3~"
  auml  = "a~H-13V2F35~H~FV-2H3~"
  Ouml  = "O~H-16V6F35~H~FV-6H3~"
  ouml  = "o~H-14V2F35~H~FV-2H3~"
  Uuml  = "U~H-15V6F35~H~FV-6H3~"
  uuml  = "u~H-13V2F35~H~FV-2H3~"
  str   = "Umlaute:    "+Auml+" "+auml+"   "+Ouml+" "+ouml+"   "+Uuml+" "+uuml
  ndcres@txFontColor            = "black"             ;-- change to black
  ndcres@txFontHeightF          =  0.025              ;-- make font size smaller
  gsn_text_ndc(wks, str, 0.09, 0.57, ndcres)

.......
```

Please turn over.




```
;-- total 43 fonts available in v6.2.0
  str_abc   = "abcdefghijklmnopqrstuvwxyz"

;-- helvetica-bold
  ndcres@txFont                 = "helvetica-bold"    ;-- change font
  ndcres@txFontColor            = "red"               ;-- change to white
  gsn_text_ndc(wks, "Font examples (43 fonts available)", 0.09, 0.50, ndcres)

;-- default: helvetica
  ndcres@txFontColor            = "black"             ;-- change to white
  ndcres@txFont                 = "helvetica"         ;-- change font
  gsn_text_ndc(wks, "helvectica :      "+str_abc, 0.09, 0.45, ndcres)

;-- times-roman
  gsn_text_ndc(wks, "~F21~times-roman :  ~F25~"+str_abc, 0.09, 0.40, ndcres)

;-- times-bold
  gsn_text_ndc(wks, "~F21~times-bold :     ~F26~"+str_abc, 0.09, 0.35, ndcres)

;-- courier
  gsn_text_ndc(wks, "~F21~courier :          ~F29~"+str_abc, 0.09, 0.30,
ndcres)
;-- courier-bold
  gsn_text_ndc(wks, "~F21~courier-bold :  ~F30~"+str_abc, 0.09, 0.25, ndcres)

;-- greek
  gsn_text_ndc(wks, "~F21~greek :            ~F33~"+str_abc, 0.09, 0.20,
ndcres)
;-- math-symbols
  str = " aeqrstu1234567890!$%&/()"
  ndcres@txFontThicknessF       =  2.0
  gsn_text_ndc(wks, "~F21~math-symbols : ~F34~"+str, 0.09, 0.15, ndcres)

;-- weather2
  gsn_text_ndc(wks, "~F21~weather2 :       ~F37~"+str_abc, 0.09, 0.10, ndcres)

;-- text-symbols
  str_abc = "bcdklmnopqrstuxyz236-+;"
  gsn_text_ndc(wks, "~F21~text-symbols :  ~F35~"+str_abc, 0.09, 0.05, ndcres)

;-- down
  down    = "write downward"
  ndcres@txFontThicknessF       =  1.0                ;-- default: 1.0
  ndcres@txFontColor            = "seagreen"          ;-- change to white
  ndcres@txJust                 = "CenterCenter"      ;-- text justification
  gsn_text_ndc(wks, "~F21D~"+down,   0.03, 0.3, ndcres)
```

6 Tickmark settings

NCL draws major and minor tickmarks at the x- and y-axis by default. In the next part the modifications of tickmarks, their spacing values and labels are demonstrated for xy- and map-plots.



6.1 XY plot

See also http://ncl.ucar.edu/Applications/tickmarks.shtml




```
;-- set minimum and maximum values of the axis
  res@trYMinF             =  -1.0          ;-- y-axis minimum value
  res@trYMaxF             =   1.0          ;-- y-axis maximum value
  res@trXMinF             =  min(x)        ;-- x-axis minimum value
  res@trXMaxF             =  max(x)        ;-- x-axis maximum value

;-- set y-axis major and minor tickmarks and tickmarks values
  res@tmYLMode            = "Manual"       ;-- set tickmarks resources manually
  res@tmYLTickSpacingF    =   0.2          ;-- label every 0.2th tickmark
  res@tmYLMinorPerMajor   =   4            ;-- draw 4 minor tickmarks between
                                           ;-- the labeled major tickmarks

;-- set x-axis major and minor tickmarks and tickmarks values
  res@tmXBMode            = "Manual"       ;-- set tickmarks resources manually
  res@tmXBTickSpacingF    =  10.0          ;-- label every 10th tickmark
  res@tmXBMinorPerMajor   =   8            ;-- draw 8 minor tickmarks between
                                           ;-- the labeled major tickmarks
```

6.2 Map plot

The default tickmark settings for a map can be changed by some **@gsn** and **@tm** resources, Not every resource can be used for other map projections than 'Cylindrical Equidistant', but on the NCL examples web page there are some examples for work arounds.

See also http://ncl.ucar.edu/Applications/mptick.shtml



Default map

```
;-- latitude settings

mpres@gsnMajorLatSpacing    =  10              ;-- change major lat tickmark spacing
mpres@gsnMinorLatSpacing    =  2.5             ;-- change major lat tickmark spacing

mpres@tmYLLabelStride       =  3               ;-- write only every 3rd label
mpres@tmYLLabelFontHeightF  =  0.016           ;-- change major lat tickmark spacing
mpres@tmYLMajorLengthF      =  0.02            ;-- change the tickmark length
mpres@tmYLMinorLengthF      =  0.01            ;-- change the tickmark length
mpres@tmYLMajorLineColor    = "blue"           ;-- change major tickmarks color
mpres@tmYLMinorLineColor    = "grey20"         ;-- change major tickmarks color
mpres@tmYLLabelFontColor    = "blue"           ;-- change label color

;-- longitude settings

mpres@gsnMajorLonSpacing    =  10              ;-- change major lon tickmark spacing
mpres@gsnMinorLonSpacing    =  2.5             ;-- change major lon tickmark spacing

mpres@tmXBLabelStride       =  4               ;-- write only every 4th label
mpres@tmXBLabelFontHeightF  =  0.014           ;-- change major lat tickmark spacing
mpres@tmXBMajorLengthF      =  0.02            ;-- change the tickmark length
mpres@tmXBMinorLengthF      =  0.01            ;-- change the tickmark length
mpres@tmYLMajorLineColor    = "red"            ;-- change major tickmarks color
mpres@tmYLMinorLineColor    = "grey20"         ;-- change major tickmarks color
mpres@tmXBLabelFontColor    = "red"            ;-- change label color

;-- grid line settings

mpres@mpGridAndLimbOn       =  True            ;-- draw grid lines on the plot
mpres@mpGridLatSpacingF     =  20              ;-- grid line lat spacing
mpres@mpGridLonSpacingF     =  45              ;-- grid line lon spacing
mpres@mpGridLineColor       = "gray"           ;-- grid line color
```



7 Legend Settings

To add a legend to a XY-plot just one resource setting has to be done:

```
res@pmLegendDisplayMode    = "Always"
```

This will add a legend below the plot with a size scaled to the plot width. To change the legend size, its position, label font, the background color or to reverse the order of the legend labels and lines the following resources are needed.



Default legend




```
res@pmLegendDisplayModce    = "Always"  ;-- display legend
res@pmLegendWidthF          =  0.18     ;-- define legend width
res@pmLegendHeightF         =  0.11     ;-- define legend height
res@pmLegendOrthogonalPosF  = -1.10     ;-- move the legend upward
res@pmLegendParallelPosF    =  0.21     ;-- move the legend to the right
```




```
res@lgAutoManage            =  False    ;-- switch auto manage off
res@lgLabelFontHeightF      =  0.022    ;-- increase label font size
```




```
res@lgPerimFill         = "SolidFill"    ;-- fill mode for legend box
res@lgPerimFillColor    = "yellow"       ;-- fill color for legend box
res@lgItemOrder         =  (/3,2,1,0/)   ;-- reverse legend
```

8 Labelbar Settings



Default labelbar

If a contour plot with fill mode on is chosen NCL will add a label bar to the plot. The label bar is centered below the plot, having nice values below the color boxes. By default no title string is written to the label bar.

The user can change the label bar width, height, position, color and font of the labels, exclude outer color boxes, add minimum and maximum labels to the outer boxes, add a title to the label bar, etc.




```
res@cnLabelBarEndStyle          = "ExcludeOuterBoxes";-- exclude the outer color boxes

res@lbTitleOn                   =  True               ;-- write title (default: "labelbar")
res@lbTitleFont                 = "courier-bold"      ;-- set title font
res@lbTitleFontColor            = "blue"              ;-- set title font
res@lbTitleFontHeightF          =  0.015              ;-- decrease the font size (default: 0.025)
res@lbTitlePosition             = "Bottom"            ;-- labelbar title postion (default: "Top")
res@lbTitleString               =  t@units            ;-- define labelbar title string
res@lbTitleOffsetF              = -0.3                ;-- move the labelbar title upwards

res@lbBoxMinorExtentF           =  0.2                ;-- decrease height of labelbar boxes and vp
res@lbBoxLinesOn                =  False              ;-- don't draw lines around labelbar boxes

res@lbLabelFontColor            = "blue"              ;-- label color
res@lbLabelFontHeightF          =  0.015              ;-- label font height
res@lbLabelFont                 = "helvetica-bold"    ;-- label font
res@lbLabelOffsetF              =  0.07               ;-- move the labelbar labels downwards

res@pmLabelBarWidthF            =  0.8                ;-- labelbar width; default is shorter
res@pmLabelBarHeightF           =  0.1                ;-- labelbar height; default is taller
res@pmLabelBarOrthogonalPosF    =  0.07               ;-- y-position (positive: downward; def: 0.02)
res@pmLabelBarParallelPosF      =  0.5                ;-- x-position (CenterCenter); default: 0.5
```

9 Insert a Logo

It is not possible to overlay a logo file onto the plot, but we can use ImageMagicks 'composite' to overlay a PNG or JPEG logo. The NCL output file type (workstation type) can be PNG, PS or PDF. The call of 'composite' can be done in the NCL script after finishing the plot using the 'system' function of NCL.




```
.........
itype = "png"           ;-- output file format NCL
otype =  itype          ;-- output file format 'composite'
.........

;---------------------------------------------------------
;-- draw the plot
;---------------------------------------------------------
  draw(plot)            ;-- create the plot
  frame(wks)            ;-- advance the frame
  delete(wks)           ;-- make sure that the output file is closed

;---------------------------------------------------------
;-- add a logo to the finished plot (upper left corner)
;-- (this could be done only for PNG plot output)
;---------------------------------------------------------

  logo = "$HOME/Pictures/DKRZ/DKRZ_Logo_mit_Text.jpg"

  if(itype .eq."pdf".or.itype.eq."ps".or.itype.eq."eps") then
     cmd  = "composite -geometry 80x80+20+20 "+logo+" plot_text_tipps."+\
            itype+" plot_text_tipps."+otype
  else
     cmd  = "composite -geometry 300x300+70+130 "+logo+" plot_text_tipps."+\
            itype+" plot_text_tipps."+otype
  end if

  system(cmd)
```

10 Change the size of the plot output file

The opening of a workstation is required prior to the creation of one or more plots. The workstation is a kind of a bin which contains all graphical instructions to build the plot. The workstation has his own resources (wk) for setting the size of the workstation, background color, foreground color, file name of the output file, height, width etc. Some resources are working only with PS, PDF or EPSI output file format like orientation or paper size.

See also http://www.ncl.ucar.edu/Document/Graphics/Resources/wk.shtml

```
;-- define a workstation

  wks_type  =  "ps"

;-- for posters, presentations or publications use a scalable or high resolution output file

  if(wks_type .eq."x11".or.wks_type.eq."png") then
     wks_type@wkWidth   = 2500
     wks_type@wkHeight  = 2500
  end if

  if(wks_type.eq."ps".or.wks_type.eq."eps".or.wks_type.eq."epsi".or.wks_type.eq."pdf") then
     wks_type@wkOrientation  = "landscape"
     wks_type@wkPaperSize    = "A3"
  end if

;-- the background color will be changed only for the plot space (1.0x1.0), but
;-- the circumfluent space for PS, PDF, EPS, EPSI will be the default background color

  wks_type@wkBackgroundColor = "yellow"
  wks_type@wkForegroundColor = "navy"

;-- open a workstation with the resource settings above

  wks  = gsn_open_wks(wks_type, "plot_wk")
```

11 Crop white space around the plot

Sometimes the white space around the plot is annoying and you want to get rid of it. Another time to use an ImageMagicks tool, called 'convert,' to deal with the output plot.

The next command line call will cut off all white space around the plot. This command string can be used with the 'system' function in a NCL script directly after finishing the plot.

```
convert -alpha off -background white -density 300 –trim <inputfile> <outputfile>
```

12 Appendix A: Font Table

A complete overview about all fonts and their key assignments can be found on the NCL web page:

http://ncl.ucar.edu/Document/Graphics/font_tables.shtml



13 Appendix B: Text Alignment

See also NCL example:

http://ncl.ucar.edu/Applications/Scripts/text_13.ncl



14 Appendix C: Example scripts

defaults.ncl

```
;----------------------------------------------------------------------
;-- NCL Script:           defaults.ncl
;--
;-- Description:          show gsn_csm plot style default settings
;--
;-- 23.06.14              K.Meier-Fleischer, meier-fleischer(at)dkrz.de
;----------------------------------------------------------------------
;load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl" ;-- not needed since v6.2.0
;load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"  ;-- not needed since v6.2.0
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/shea_util.ncl"

begin

;-- generate dummy data (must be of same dimension size) for the xy-plots
  x              =  ispan(0,59,1)                ;-- x-array from 0 to 60
  x@long_name    = "x variable long_name"        ;-- define long_name attribute

  data           =  random_uniform(-3,6,60)      ;-- generate random data values
  data@long_name = "data variable long_name"     ;-- define long_name attribute
  data@units     = "data variable units"         ;-- define units attribute

;-- open a workstation
  wks  = gsn_open_wks("png", "plot_defaults")  ;-- open a workstation

;++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
;-- xy defaults
;++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
;----------------------------------------------------------
;-- 1. plot  - only defaults
;----------------------------------------------------------
  plot = gsn_csm_xy(wks, x, data, True)                          ;-- create the second plot


;-- set plot resources
  res  =  True                                                   ;-- open a kind of container to store all plot settings
  res@xyLineThicknessF           =  2.0                          ;-- thicker line
```

```
;-----------------------------------------------------------
;-- top left string (default: variable long_name attribute)
;-----------------------------------------------------------
  res@gsnLeftString             = "gsnLeftString"               ;-- overwrite top left string

;-----------------------------------------------------------
;-- top center string (defaulft: "")
;-----------------------------------------------------------
  res@gsnCenterString           = "gsnCenterString"             ;-- overwrite top center string

;-----------------------------------------------------------
;-- top right string (default: variable units attribute)
;-----------------------------------------------------------
  res@gsnRightString            = "gsnRightString"              ;-- overwrite top right string

;-----------------------------------------------------------
;-- x-axis title string settings
;-----------------------------------------------------------
  res@tiXAxisString             = "tiXAxisString"               ;-- set x-axis title string

;-----------------------------------------------------------
;-- y-axis title string settings
;-----------------------------------------------------------
  res@tiYAxisString             = "tiYAxisString"               ;-- set y-axis title string

;-----------------------------------------------------------
;-- main title string settings
;-----------------------------------------------------------
  res@tiMainString             = "tiMainString"                 ;-- set title string

;-----------------------------------------------------------
;-- 2. plot
;-----------------------------------------------------------
  plot = gsn_csm_xy(wks, x, data, res)                          ;-- create the first plot

;-----------------------------------------------------------
;-- resize the x- and y-axis length using the viewport settings
;-----------------------------------------------------------
  res@vpXF                      =  0.15                         ;-- viewport start x-position
  res@vpYF                      =  0.7                          ;-- viewport start y-position
  res@vpHeightF                 =  0.4                          ;-- viewport height
  res@vpWidthF                  =  0.8                          ;-- viewport width
```

```
;---------------------------------------------------------
;-- draw auxiliary grid lines
;---------------------------------------------------------
  drawNDCGrid(wks)                                          ;-- draw the NDC grid

;---------------------------------------------------------
;-- 3. plot
;---------------------------------------------------------
  plot = gsn_csm_xy(wks, x, data, res)                      ;-- create the second plot

;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
;-- contour, contour map defaults
;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
;-- read file and set variable
  f = addfile("$HOME/NCL/EGU/2014/Light_hands-on/ECHAM5_OM_A1B_t_20010101.nc","r") ;-- open file
  var = f->t(0,0,:,:)                                       ;-- define variable

;---------------------------------------------------------
;-- 4. plot - contour defaults
;---------------------------------------------------------
  map = gsn_csm_contour(wks,var,True)                       ;-- create the third plot

;---------------------------------------------------------
;-- 5. plot - contour map defaults
;---------------------------------------------------------
  map = gsn_csm_contour_map(wks,var,True)                   ;-- create the third plot

;---------------------------------------------------------
;-- 6. plot - contour map fill mode on
;---------------------------------------------------------
  mres             = True
  mres@cnFillOn    =  True

  map = gsn_csm_contour_map(wks,var,mres)                   ;-- create the third plot

;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
;-- vector, vector map defaults
;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
  f   =  addfile("$NCL_TUT/data/ECHAM5_OM_A1B_2001_0101-1001_2D.nc","r")
  u   =  f->u10(0,:,:)                          ;-- u-velocity, first time step
  v   =  f->v10(0,:,:)                          ;-- v-velocity, first time step
```

```
;----------------------------------------------------------
;-- 7. plot - vector defaults
;----------------------------------------------------------
  vector = gsn_csm_vector(wks,u,v,True)

;----------------------------------------------------------
;-- 8. plot - vector map defaults
;----------------------------------------------------------
  vector = gsn_csm_vector_map(wks,u,v,True)

end
```

titles.ncl:

```
;----------------------------------------------------------------------
;-- NCL Script:           titles_contour.ncl
;--
;-- Description:          show title string capability
;--
;-- 25.06.14              K.Meier-Fleischer, meier-fleischer(at)dkrz.de
;----------------------------------------------------------------------
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl" ;-- not needed since v6.2.0
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"  ;-- not needed since v6.2.0

begin

;-- read file and set variable
  f = addfile("$HOME/NCL/EGU/2014/Light_hands-on/ECHAM5_OM_A1B_t_20010101.nc","r")     ;-- open file
  t = f->t(0,0,:,:)                                    ;-- define variable

;-- open a workstation
  wks_type          = "png"
  wks_type@wkWidth  = 1500
  wks_type@wkHeight = 1500

  wks  = gsn_open_wks(wks_type, "plot_titles_contour")  ;-- open a workstation

;-- set plot resources
  res  =  True                 ;-- open a kind of container to store all plot settings
```

```
;-----------------------------------------------------------
;-- top left string (default: variable long_name attribute)
;-----------------------------------------------------------
  res@gsnLeftStringFontHeightF   =  0.008           ;-- set gsnLeftString font size
  res@gsnLeftStringFontColor     = "red"             ;-- set top left string color

;-----------------------------------------------------------
;-- top center string (defaulft: "")
;-----------------------------------------------------------
  res@gsnCenterString            = "gsnCenterString" ;-- overwrite top center string
  res@gsnCenterStringFontHeightF =  0.013             ;-- set gsnCenterString font size
  res@gsnCenterStringFontColor   = "green"           ;-- set top center string color

;-----------------------------------------------------------
;-- top right string (default: variable units attribute)
;-----------------------------------------------------------
  res@gsnRightStringFontHeightF  =  0.015           ;-- set gsnLeftString font size
  res@gsnRightStringFontColor    = "darkorchid"      ;-- set top right string color

  res@gsnLeftStringParallelPosF     = -0.05              ;-- move the LeftString slightly left
  res@gsnRightStringParallelPosF    =  1.01              ;-- move the RightString slightly right
                                                        ;--(moves also the tiXAxisString if set to "right")
;-----------------------------------------------------------
;-- common font setting for gsn- and ti-resources
;-----------------------------------------------------------
  res@gsnStringFont              = "courier"         ;-- set gsnLeft/Center/RightString font
  res@tiMainFont                 = "times-bold"      ;-- set main title font

;-----------------------------------------------------------
;-- main title string settings
;-----------------------------------------------------------
  res@tiMainString               = "This is a title line~C~~Z50~         with a continuing line"    ;-- overwrite top
left string

  res@tiMainFontHeightF          =  0.05             ;-- change main title font size
  res@tiMainFontColor            = "blue"            ;-- set main title font color

;-----------------------------------------------------------
;-- x-axis title string settings
;-----------------------------------------------------------
  res@tiXAxisString              = "tiXAxisString"   ;-- set x-axis title string
```

```
  res@tiXAxisFontHeightF         =  0.02             ;-- change x-axis font size
  res@tiXAxisFontColor           = "red"             ;-- set x-axis title color

;---------------------------------------------------------
;-- y-axis title string settings
;---------------------------------------------------------
  res@tiYAxisString              = "tiYAxisString"   ;-- set y-axis title string

  res@tiYAxisFontHeightF         =  0.02             ;-- change y-axis font size
  res@tiYAxisFontColor           = "navy"            ;-- set y-axis title color

  res@tiYAxisSide                = "right"           ;-- draw x-axis title string on the left of plot
  res@tiYAxisAngleF              = 270               ;-- rotate the Y-axis title 270 degrees

;---------------------------------------------------------
;-- labelbar settings
;---------------------------------------------------------
  res@lbTitleOn                  =  True              ;-- write title (default: "labelbar")
  res@lbTitleFontHeightF         =  0.014             ;-- decrease the font size (default: 0.025)
  res@lbTitlePosition            = "Bottom"           ;-- labelbar title postion (default: "Top")
  res@lbTitleString              = "Temperature in Kelvin" ;-- define labelbar title string

;---------------------------------------------------------
;-- create the plot
;---------------------------------------------------------
  res@cnFillOn                   =  True
  res@cnLinesOn                  =  False

  plot = gsn_csm_contour_map(wks, t, res)            ;-- create the default plot

end
```

additional_text.ncl:

```
;----------------------------------------------------------------------
;-- NCL Script:           additional_text.ncl
;--
;-- Description:          show text capability
;--
;-- 25.06.14              K.Meier-Fleischer, meier-fleischer(at)dkrz.de
;----------------------------------------------------------------------
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl" ;-- not needed since v6.2.0
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"  ;-- not needed since v6.2.0

begin

;-- generate dummy data

  x  = ispan(0,29,1)
  y  = random_uniform(-3,6,30)


;-- open a workstation
  wks  = gsn_open_wks("png", "plot_add_text")        ;-- open a workstation

;-- set plot resources
  res  =  True                 ;-- open a kind of container to store all plot settings

  res@gsnDraw                   =  False            ;-- don't draw plot yet
  res@gsnFrame                  =  False            ;-- don't advance frame

  res@xyLineThicknessF          =  2.0              ;-- increase the line width

;---------------------------------------------------------
;-- create the plot
;---------------------------------------------------------
  plot = gsn_csm_xy(wks, x, y, res)                  ;-- create the default xy-plot

;---------------------------------------------------------
;-- additional text on plot using plot coordinate
;---------------------------------------------------------
  txres                         =  True             ;-- text resources additional text
  txres@txFontColor             = "blue"            ;-- change to white
  txres@txFontHeightF           =  0.03             ;-- decrease font size
```

```
  txres@txJust                   = "CenterCenter"     ;-- text justification

  id = gsn_add_text(wks, plot, "gsn_add_text", 14, 4, txres) ;-- center position x=14, y=4

  txres@txFontColor              = "green"            ;-- change to white
  txres@txFontHeightF            =  0.02              ;-- decrease font size
  txres@txAngleF                 = 45                 ;-- rotate text counterclockwise 45 deg.

  id = gsn_add_text(wks, plot, "gsn_add_text rotate 45~S~o~N~", 22, -1, txres) ;-- center position x=22, y=-1

  txres@txFontColor              = "navy"             ;-- change to white
  txres@txFontHeightF            =  0.02              ;-- decrease font size
  txres@txAngleF                 = -45                ;-- rotate text counterclockwise 45 deg.
  txres@txBackgroundFillColor    = "yellow"           ;-- text backround color

  id = gsn_add_text(wks, plot, "gsn_add_text bg", 8, 0, txres) ;-- center position x=8, y=0

  txres@txFontColor              = "red"              ;-- change to white
  txres@txFontHeightF            =  0.035              ;-- decrease font size
  txres@txFontOpacityF           =  0.5               ;-- set transparency of text
  txres@txJust                   = "BottomLeft"       ;-- text justification
  txres@txBackgroundFillColor    = -1                 ;-- set text background color to fully transparent
  delete(txres@txAngleF)                              ;-- reset text rotation

  id = gsn_add_text(wks, plot, "gsn_add_text transparent", 1, 2, txres) ;-- center position x=1, y=2

  value = 3.83927489235
  str = "formatted value "+sprintf("%3.4f", value)+" in text"

  txres@txFontColor              = "violet"           ;-- change to white
  txres@txFontOpacityF           =  1.0               ;-- set fully opacity of text
  txres@txFontHeightF            =  0.02              ;-- decrease font size

  id  = gsn_add_text(wks, plot, str, 1, -3.5, txres)

  draw(plot)
  frame(wks)

;---------------------------------------------------------
;-- 2. plot
;---------------------------------------------------------
  plot = gsn_csm_xy(wks, x, y, res)                   ;-- create the default xy-plot
```

```
;---------------------------------------------------------
;-- additional text on 2. plot using page coordinate (NDC)
;---------------------------------------------------------
  ndcres                        =  True              ;-- text resources copyright string

;-- text strings on left top of the plot
  ndcres@txFontHeightF           =  0.03              ;-- decrease font size
  ndcres@txJust                  = "BottomLeft"       ;-- text justification

  line1 = "Text line 1"
  line2 = "Text line 2"
  line3 = "Text line 3"

  gsn_text_ndc(wks, line1, 0.15, 0.95, ndcres)
  ndcres@txFontHeightF           =  0.015             ;-- decrease font size
  gsn_text_ndc(wks, line2, 0.15, 0.90, ndcres)
  ndcres@txFontHeightF           =  0.01              ;-- decrease font size
  gsn_text_ndc(wks, line3, 0.15, 0.87, ndcres)

;-- user defined axis title strings
  ndcres@txFontHeightF          =  0.018             ;-- decrease font size
  ndcres@txFontColor            = "blue"             ;-- change to white
  ndcres@txAngleF               =  90                ;-- rotate text counterclockwise 90 deg.
  ndcres@txJust                 = "CenterCenter"     ;-- text justification

  gsn_text_ndc(wks, "y-axis title string", 0.09, 0.5, ndcres)
  ndcres@txAngleF               =  -90               ;-- rotate text clockwise 90 deg.
  gsn_text_ndc(wks, "y-axis title string", 0.85, 0.5, ndcres)

  ndcres@txAngleF               =  0                 ;-- unrotate text
;  delete(ndcres@txAngleF)
  gsn_text_ndc(wks, "x-axis title string", 0.5, 0.12, ndcres)
  gsn_text_ndc(wks, "x-axis title string", 0.5, 0.83, ndcres)


;-- super script
  ndcres@txFontHeightF          =  0.04              ;-- make font size smaller
  ndcres@txFontColor            = "orange"           ;-- change to white
  ndcres@txJust                 = "BottomLeft"       ;-- text justification

  super = "35.5~S~o~N~C"
  sub   = "H~B~2~N~O"
```

```
  gsn_text_ndc(wks, super, 0.23, 0.75, ndcres)

  ndcres@txAngleF               = 45                ;-- rotate text counterclockwise 45 deg.
  gsn_text_ndc(wks, sub,   0.7, 0.25, ndcres)

  delete(ndcres@txAngleF)

;-- Umlaute
  Auml    = "A~H-15V6F35~H~FV-6H3~"
  auml    = "a~H-13V2F35~H~FV-2H3~"
  Ouml    = "O~H-16V6F35~H~FV-6H3~"
  ouml    = "o~H-14V2F35~H~FV-2H3~"
  Uuml    = "U~H-15V6F35~H~FV-6H3~"
  uuml    = "u~H-13V2F35~H~FV-2H3~"

  ndcres@txFontColor            = "black"           ;-- change to white
  ndcres@txFontHeightF          =  0.025            ;-- make font size smaller

  gsn_text_ndc(wks, "Umlaute: "+Auml+" "+auml+"   "+Ouml+" "+ouml+"   "+Uuml+" "+uuml, 0.05, 0.06, ndcres)

;-- copyright
  ndcres@txFontColor            = "red"             ;-- change to white
  ndcres@txFontHeightF          =  0.013            ;-- make font size smaller
  ndcres@txJust                 = "BottomRight"     ;-- text justification

  gsn_text_ndc(wks,"gsn_text_ndc  -->  ~F35~c ~F21~~N~DKRZ 2014", 0.9, 0.02, ndcres)
                                                    ;-- plot copyright info
;---------------------------------------------------------
;-- draw the plot
;---------------------------------------------------------

  draw(plot)
  frame(wks)

end
```

function_codes.ncl:

```
;----------------------------------------------------------------------
;-- NCL Script:           function_codes.ncl
;--
;-- Description:          show text function codes
;--
;-- 30.06.14              K.Meier-Fleischer, meier-fleischer(at)dkrz.de
;----------------------------------------------------------------------
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl" ;-- not needed since v6.2.0
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"  ;-- not needed since v6.2.0

begin

;-- open a workstation
  wks  = gsn_open_wks("png", "plot_function_codes")  ;-- open a workstation

;-- set plot resources
  ndcres  =  True     ;-- open a kind of container to store all plot settings
  ndcres@gsnDraw                =  False             ;-- don't advance frame
  ndcres@gsnFrame               =  False             ;-- don't advance frame

  ndcres@txFontHeightF          =  0.03              ;-- decrease font size
  ndcres@txJust                 = "CenterCenter"     ;-- text justification
  ndcres@txFont                 = "helvetica-bold"   ;-- change font

  txid = gsn_create_text_ndc(wks,"Fonts and Function Codes", 0.5,0.96,ndcres)

  ndcres@txFontHeightF          =  0.03              ;-- decrease font size
  ndcres@txJust                 = "BottomLeft"       ;-- text justification
  ndcres@txFont                 = "helvetica"        ;-- change font

  line1 = "Font Height = 0.03"
  gsn_text_ndc(wks, line1, 0.09, 0.86, ndcres)

  line2 = "Font Height = 0.015"
  ndcres@txFontHeightF          =  0.015             ;-- decrease font size
  gsn_text_ndc(wks, line2, 0.09, 0.83, ndcres)

  line3 = "Font Height = 0.01"
  ndcres@txFontHeightF          =  0.01              ;-- decrease font size
  gsn_text_ndc(wks, line3, 0.09, 0.80, ndcres)
```

```
;-- superscripting
  ndcres@txFontHeightF          =  0.03              ;-- make font size smaller
  ndcres@txFontColor            = "blue"             ;-- change to white
  ndcres@txJust                 = "BottomLeft"       ;-- text justification

  super = "Superscripting:  35.5~S~o~N~   x~S~2~N~   kgm~S~-2~N~s~S~-1~N~"
  gsn_text_ndc(wks, super, 0.09, 0.74, ndcres)

;-- subscripting
  sub   = "Subscripting:     H~B~2~N~O   CO~B~2~N~"
  gsn_text_ndc(wks, sub,   0.09, 0.69, ndcres)

;-- different font size in one string
  str_one = "Different ~Z70~font ~Z130~sizes~Z100~ in one ~Z60~string~Z100~"
  ndcres@txFontColor            = "seagreen"             ;-- change to white
  gsn_text_ndc(wks, str_one, 0.09, 0.63, ndcres)

;-- Umlaute
  Auml  = "A~H-15V6F35~H~FV-6H3~"
  auml  = "a~H-13V2F35~H~FV-2H3~"
  Ouml  = "O~H-16V6F35~H~FV-6H3~"
  ouml  = "o~H-14V2F35~H~FV-2H3~"
  Uuml  = "U~H-15V6F35~H~FV-6H3~"
  uuml  = "u~H-13V2F35~H~FV-2H3~"
  str   = "Umlaute:    "+Auml+" "+auml+"   "+Ouml+" "+ouml+"   "+Uuml+" "+uuml

  ndcres@txFontColor            = "black"            ;-- change to white
  ndcres@txFontHeightF          =  0.025             ;-- make font size smaller
  gsn_text_ndc(wks, str, 0.09, 0.57, ndcres)

;-- total 43 fonts available in v6.2.0

  str_abc   = "abcdefghijklmnopqrstuvwxyz"

;-- helvetica-bold
  ndcres@txFont                 = "helvetica-bold"   ;-- change font
  ndcres@txFontColor            = "red"              ;-- change to white
  gsn_text_ndc(wks, "Font examples (43 fonts available in NCL)", 0.09, 0.50, ndcres)

;-- default: helvetica
  ndcres@txFontColor            = "black"            ;-- change to white
  ndcres@txFont                 = "helvetica"        ;-- change font
```

```
  gsn_text_ndc(wks, "helvectica :      "+str_abc, 0.09, 0.45, ndcres)

;-- times-roman
;  ndcres@txFont                 = "times-roman"     ;-- change font
  gsn_text_ndc(wks, "~F21~times-roman :  ~F25~"+str_abc, 0.09, 0.40, ndcres)

;-- times-bold
;  ndcres@txFont                 = "times-bold"      ;-- change font
  gsn_text_ndc(wks, "~F21~times-bold :     ~F26~"+str_abc, 0.09, 0.35, ndcres)

;-- courier
;  ndcres@txFont                 = "courier"         ;-- change font
  gsn_text_ndc(wks, "~F21~courier :          ~F29~"+str_abc, 0.09, 0.30, ndcres)

;-- courier-bold
;  ndcres@txFont                 = "courier-bold"    ;-- change font
  gsn_text_ndc(wks, "~F21~courier-bold :  ~F30~"+str_abc, 0.09, 0.25, ndcres)

;-- greek
;  ndcres@txFont                 = "greek"           ;-- change font
  gsn_text_ndc(wks, "~F21~greek :            ~F33~"+str_abc, 0.09, 0.20, ndcres)

;-- math-symbols
  ndcres@txFontThicknessF       =  2.0
;  ndcres@txFont                 = "math-symbols"    ;-- change font
  gsn_text_ndc(wks, "~F21~math-symbols : ~F34~aeqrstu1234567890!$%&/()", 0.09, 0.15, ndcres)

;-- weather2
;  ndcres@txFont                 = "weather2"        ;-- change font
  gsn_text_ndc(wks, "~F21~weather2 :       ~F37~"+str_abc, 0.09, 0.10, ndcres)

;-- text-symbols
  str_abc = "bcdklmnopqrstuxyz236-+;"
;  ndcres@txFont                 = "text-symbols"    ;-- change font
  gsn_text_ndc(wks, "~F21~text-symbols :  ~F35~"+str_abc, 0.09, 0.05, ndcres)

;-- across / down
  down   = "write downward"
  ndcres@txFontThicknessF       =  1.0              ;-- default: 1.0
  ndcres@txFontColor            = "seagreen"        ;-- change to white
  ndcres@txJust                 = "CenterCenter"    ;-- text justification
  gsn_text_ndc(wks, "~F21D~"+down,   0.03, 0.3, ndcres)
```

```
;-- create the plot
  draw(txid)
  frame(wks)
end
```

tickmarks.ncl:

```
;----------------------------------------------------------------------
;-- NCL Script:           tickmarks.ncl
;--
;-- Description:          tickmarks settings
;--                         - xy plot
;--                         - map plot
;--
;-- 01.07.14              K.Meier-Fleischer, meier-fleischer(at)dkrz.de
;----------------------------------------------------------------------
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl" ;-- not needed since v6.2.0
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"  ;-- not needed since v6.2.0

begin

;-- generate dummy data
  x  = ispan(0,100,1)
  y  = cos(0.0628*ispan(0,100,1))              ;-- generate a curve with 101 points.

;-- open a workstation with the resource settings above
  wks = gsn_open_wks("pdf", "plot_tickmarks")

;-- resource settings
  res                   =  True

;-- define the minimum and maximum values of the axis
  res@trXMinF           =  min(x)              ;-- x-axis minimum value
  res@trXMaxF           =  max(x)              ;-- x-axis maximum value
  res@trYMinF           =  -1.0                ;-- y-axis minimum value
  res@trYMaxF           =   1.0                ;-- y-axis maximum value

;-- set x-axis major and minor tickmarks and tickmarks values
  res@tmXBMode          = "Manual"             ;-- set tickmarks resources manually
  res@tmXBTickSpacingF  =  10.0                ;-- label every 10th tickmark
```

```
  res@tmXBMinorPerMajor =   8                   ;-- draw 8 minor tickmarks between the labeled major tickmarks
;-- set y-axis major and minor tickmarks and tickmarks values
  res@tmYLMode          = "Manual"              ;-- set tickmarks resources manually
  res@tmYLTickSpacingF  =   0.2                 ;-- label every 0.2th tickmark
  res@tmYLMinorPerMajor =   4                   ;-- draw 4 minor tickmarks between the labeled major tickmarks

;-- create the plot
  plot = gsn_csm_xy(wks,x,y,res)

;-- map plot
  f = addfile("$NCL_TUT/data/ECHAM5_OM_A1B_2001_0101-1001_2D.nc","r") ;-- open data file
  var = f->tsurf(0,:,:)                         ;-- read variable

;-- resource settings
  mpres                       =  True

;-- create default map plot
  plot = gsn_csm_contour_map(wks,var,mpres)

;-- latitude settings
  mpres@gsnMajorLatSpacing   =  10              ;-- change major lat tickmark spacing
  mpres@gsnMinorLatSpacing   =  2.5             ;-- change major lat tickmark spacing

  mpres@tmYLLabelStride      =  3               ;-- write only every 3rd label
  mpres@tmYLLabelFontHeightF =  0.016           ;-- change major lat tickmark spacing
  mpres@tmYLMajorLengthF     =  0.02            ;-- change the tickmark length
  mpres@tmYLMinorLengthF     =  0.01            ;-- change the tickmark length
  mpres@tmYLMajorLineColor   = "blue"           ;-- change major tickmarks color
  mpres@tmYLMinorLineColor   = "grey20"         ;-- change major tickmarks color
  mpres@tmYLLabelFontColor   = "blue"           ;-- change label color

;-- longitude settings
  mpres@gsnMajorLonSpacing   =  10              ;-- change major lon tickmark spacing
  mpres@gsnMinorLonSpacing   =  2.5             ;-- change major lon tickmark spacing

  mpres@tmXBLabelStride      =  4               ;-- write only every 4th label
  mpres@tmXBLabelFontHeightF =  0.014           ;-- change major lat tickmark spacing
  mpres@tmXBMajorLengthF     =  0.02            ;-- change the tickmark length
  mpres@tmXBMinorLengthF     =  0.01            ;-- change the tickmark length
  mpres@tmXBMajorLineColor   = "red"            ;-- change major tickmarks color
  mpres@tmXBMinorLineColor   = "grey20"         ;-- change major tickmarks color
  mpres@tmXBLabelFontColor   = "red"            ;-- change label color
```

```
;-- grid line settings
  mpres@mpGridAndLimbOn      = True            ;-- draw grid lines on the plot
  mpres@mpGridLatSpacingF    = 20              ;-- grid line lat spacing
  mpres@mpGridLonSpacingF    = 45              ;-- grid line lon spacing
  mpres@mpGridLineColor      = "gray"          ;-- grid line color

;-- create the plot
  plot = gsn_csm_contour_map(wks,var,mpres)

end
```

legend.ncl:

```
;----------------------------------------------------------------------
;-- NCL Script:           legend_settings.ncl
;--
;-- Description:          show legend defaults and resource settings
;--
;-- 10.07.14              K.Meier-Fleischer, meier-fleischer(at)dkrz.de
;----------------------------------------------------------------------
;load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl" ;-- not needed since v6.2.0
;load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"  ;-- not needed since v6.2.0

begin
;-- generate dummy data (must be of same dimension size) for the xy-plots
  dims           = 50

  x              =  ispan(0,dims-1,1)                    ;-- x-array from 0 to 60
  x@long_name    = "x variable long_name"                ;-- define long_name attribute

  data           =  new((/4,dims/),float)
  data(0,:)      =  random_uniform(0.,1600.,dims)        ;-- generate random data values
  data(1,:)      =  random_uniform(0.,1600.,dims)        ;-- generate random data values
  data(2,:)      =  random_uniform(0.,1600.,dims)        ;-- generate random data values
  data(3,:)      =  random_uniform(0.,1600.,dims)        ;-- generate random data values
  data@long_name = "data variable long_name"             ;-- define long_name attribute
  data@units     = "data variable units"                 ;-- define units attribute

;-- open a workstation
```

```
  wks   = gsn_open_wks ("png","plot_legend_settings")

;-- set resources
  res                          =  True
  res@gsnMaximize              =  True                        ;-- maximize output plot

  res@tiXAxisString            = "velocity in m/s"            ;-- x-axis title
  res@tiYAxisString            = "height in m"                ;-- y-axis title

  res@trYMinF                  =  -100                        ;-- y-axis minimum value
  res@trYMaxF                  =  2000                        ;-- y-axis maximum value

  res@xyLineColors             =  (/"red","blue","black","green"/) ;-- set the line colors.
  res@xyDashPatterns           =      (/0,1,1,0/)                   ;-- line style
  res@xyLineLabelFontColor     =      1                               ;-- black line label color
  res@xyLineThicknessF         =  3.0                         ;-- line thickness for all lines
  res@xyExplicitLegendLabels = (/"u-inst", "u-stat", "v-inst", "v-stat"/)

  res@pmLegendDisplayMode    = "Always"                       ;-- display legend

;-- create 1. plot
  plot  = gsn_xy (wks,x,data,res)

;-- 2. plot define legend box size and move it upwards into the plot
  res@pmLegendDisplayMode    = "Always"                       ;-- display legend
  res@pmLegendWidthF         =  0.18                          ;-- define legend width
  res@pmLegendHeightF        =  0.11                          ;-- define legend height
  res@pmLegendOrthogonalPosF = -1.10                          ;-- move the legend upward
  res@pmLegendParallelPosF   =  0.21                          ;-- move the legend to the right

;-- create 2. plot
  plot  = gsn_xy (wks,x,data,res)

;-- 3. plot increase legend label font, increase space between legend lines and labels

  res@xyExplicitLegendLabels = (/"  u-inst", "  u-stat", "  v-inst", "  v-stat"/) ;-- increase space between line
                                                                                  ;-- and label
  res@pmLegendDisplayMode    = "Always"                       ;-- display legend
  res@pmLegendWidthF         =  0.18                          ;-- define legend width
  res@pmLegendHeightF        =  0.11                          ;-- define legend height
  res@pmLegendOrthogonalPosF = -1.10                          ;-- move the legend upward
  res@pmLegendParallelPosF   =  0.21                          ;-- move the legend to the right
```

```
  res@lgAutoManage           =  False                     ;-- switch auto manage off
  res@lgLabelFontHeightF     =  0.022                     ;-- increase label font size

;-- create 3. plot
  plot  = gsn_xy (wks,x,data,res)

;-- 4. plot reverse lines and labels, legend box fill color yellow

  res@xyExplicitLegendLabels = (/"  u-inst", "  u-stat", "  v-inst", "  v-stat"/) ;-- increase space between line
                                                                                  ;-- and label
  res@pmLegendWidthF         =  0.18                      ;-- define legend width
  res@pmLegendHeightF        =  0.11                      ;-- define legend height
  res@pmLegendOrthogonalPosF = -1.10                      ;-- move the legend upward
  res@pmLegendParallelPosF   =  0.21                      ;-- move the legend to the right

  res@lgAutoManage           =  False                     ;-- switch auto manage off
  res@lgLabelFontHeightF     =  0.022                     ;-- increase label font size
;  res@lgPerimOn               =  True                     ;-- no box around
  res@lgPerimFill            = "SolidFill"                ;-- fill mode for legend box
  res@lgPerimFillColor       = "yellow"                   ;-- fill color for legend box
  res@lgItemOrder            =  (/3,2,1,0/)               ;-- reverse legend

;-- create 4. plot
  plot  = gsn_xy (wks,x,data,res)

end
```

labelbar.ncl:

```
;----------------------------------------------------------------------
;-- NCL Script:           labelbar_settings.ncl
;--
;-- Description:          change labelbar default settings
;--
;-- 02.07.14              K.Meier-Fleischer, meier-fleischer(at)dkrz.de
;----------------------------------------------------------------------
;load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl" ;-- not needed since v6.2.0
;load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"  ;-- not needed since v6.2.0
```

```
begin

;-- read file and set variable
  f = addfile("$HOME/NCL/EGU/2014/Light_hands-on/ECHAM5_OM_A1B_t_20010101.nc","r")     ;-- open file
  t = f->t(0,0,:,:)                                    ;-- define variable

;-- open a workstation
  wks  = gsn_open_wks("pdf", "plot_labelbar")          ;-- open a workstation
  gsn_define_colormap(wks,"ncl_default")

;---------------------------------------------------------
;-- default plot
;---------------------------------------------------------
  cnres                          =  True               ;-- open a kind of container to store all plot settings
  cnres@cnFillOn                 =  True               ;-- set fill mode on

  plot = gsn_csm_contour_map(wks,t,cnres)

;---------------------------------------------------------
;-- labelbar modifications
;---------------------------------------------------------

;-- set plot resources
  res                            =  True               ;-- open a kind of container to store all plot settings
  res@cnFillOn                   =  True               ;-- set fill mode on

;---------------------------------------------------------
;-- labelbar settings
;---------------------------------------------------------
  res@cnLabelBarEndStyle         = "ExcludeOuterBoxes";-- exclude the outer color boxes

  res@lbTitleOn                  =  True                ;-- write title (default: "labelbar")
  res@lbTitleFont                = "courier-bold"       ;-- set title font
  res@lbTitleFontColor           = "blue"               ;-- set title font
  res@lbTitleFontHeightF         =  0.015               ;-- decrease the font size (default: 0.025)
  res@lbTitlePosition            = "Bottom"             ;-- labelbar title postion (default: "Top")
  res@lbTitleString              =  t@units             ;-- define labelbar title string
  res@lbTitleOffsetF             = -0.3                 ;-- move the labelbar title upwards

  res@lbBoxMinorExtentF          =  0.2                 ;-- decrease height of labelbar boxes
  res@lbBoxLinesOn               =  False               ;-- no lines around labelbar boxes
```

```
;  res@lbLabelStride               =  2                 ;-- skip every other label
  res@lbLabelFontColor           = "blue"              ;-- label color
;  res@lbLabelPosition             = "Bottom"           ;-- where to write the labelbar title
;  res@lbLabelAlignment            = "InteriorEdges"    ;-- where to write the labels
  res@lbLabelFontHeightF         =  0.015              ;-- label font height
  res@lbLabelFont                = "helvetica-bold"    ;-- label font
  res@lbLabelOffsetF             =  0.07               ;-- move the labelbar labels downwards

  res@pmLabelBarWidthF           =  0.8                ;-- labelbar width; default is shorter
  res@pmLabelBarHeightF          =  0.1                ;-- labelbar height; default is taller
  res@pmLabelBarOrthogonalPosF   =  0.07               ;-- y-position (positive: downward); default: 0.02
  res@pmLabelBarParallelPosF     =  0.5                ;-- x-position (CenterCenter); default: 0.5

;---------------------------------------------------------
;-- 2. plot
;---------------------------------------------------------

  plot = gsn_csm_contour_map(wks, t, res)            ;-- create the default plot

end
```

text_tipps.ncl:

```
;----------------------------------------------------------------------
;-- NCL Script:           text_tipps.ncl
;--
;-- Description:          show text capability + insert a logo
;--
;-- 06.06.14              K.Meier-Fleischer, meier-fleischer(at)dkrz.de
;----------------------------------------------------------------------
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl" ;-- not needed since v6.2.0
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"  ;-- not needed since v6.2.0

begin

;-- generate dummy data

  x  = ispan(0,100,1)
  y  = cos(0.0628*ispan(0,100,1))                    ;-- generate a curve with 101 points.
```

```
;-- output type
  itype = "png"           ;-- wks type and input format for composite call
  otype =  itype          ;-- output file format for composite call

;-- open a workstation
  wks_type   =  itype

  if(itype .eq."png") then

     wks_type@wkWidth   = 2500
     wks_type@wkHeight  = 2500
  end if

  wks  = gsn_open_wks(wks_type, "plot_text_tipps")

;-- set plot resources

  res  =  True                 ;-- open a kind of container to store all plot settings
  res@gsnDraw                    =  False            ;-- don't draw plot yet
  res@gsnFrame                   =  False            ;-- don't advance frame
  res@gsnYRefLine                =  0.0              ;-- create a reference line
  res@gsnYRefLineThicknessF      =  4.0              ;-- create a reference line

  res@xyLineThicknessF           =  4.0              ;-- line thickness

  res@tiMainString               = "          Cosinus function~C~y = cos(0.0628*ispan(0,100,1))" ;-- main title string
  res@tiXAxisString              = "x-axis"          ;-- set x-axis title string
  res@tiYAxisString              = "y-axis"          ;-- set x-axis title string
  res@tiMainOffsetYF             =  0.05
  res@tiMainFont                 = "times-roman"     ;-- set gsnLeft/Center/RightString font

  res@tmXBLabelFontHeightF       =  0.025            ;-- increase x-axis label font size
  res@tmYLLabelFontHeightF       =  0.025            ;-- increase y-axis label font size

;---------------------------------------------------------
;-- create the plot
;---------------------------------------------------------
  plot = gsn_csm_xy(wks, x, y, res)                  ;-- create the default plot

;---------------------------------------------------------
;-- additional text on plot using plot coordinate
;---------------------------------------------------------
```

```
  txres                          =  True              ;-- text resources additional text
  txres@txFontColor              = "blue"             ;-- change to white
  txres@txFontHeightF            =  0.015             ;-- decrease font size
  txres@txJust                   = "CenterCenter"     ;-- text justification

  id = gsn_add_text(wks, plot, "maximum", 50, 0.99, txres)  ;-- center position x=50, y=0.99
  id = gsn_add_text(wks, plot, "minimum", 10, -0.99, txres) ;-- center position x=10, y=-0.99

;---------------------------------------------------------
;-- draw red lines
;---------------------------------------------------------
  plres  =  True
  plres@gsLineColor              = "red"
  plres@gsLineThicknessF         =  4.0

  plid1 = gsn_add_polyline(wks, plot, (/ 5,40/), (/0.99,0.99/), plres)
  plid2 = gsn_add_polyline(wks, plot, (/60,95/), (/0.99,0.99/), plres)
  plid3 = gsn_add_polyline(wks, plot, (/20,45/), (/-0.99,-0.99/), plres)

;---------------------------------------------------------
;-- additional text on plot using page coordinate (NDC)
;---------------------------------------------------------
  ndcres                         =  True              ;-- text resources copyright string
  ndcres@txFontColor             = "green"            ;-- change to white
  ndcres@txFontHeightF           =  0.02              ;-- make font size smaller
  ndcres@txJust                  = "CenterCenter"     ;-- text justification

  gsn_text_ndc(wks,"x-axis", 0.5, 0.83, ndcres)      ;-- draw a axis label on the right

  ndcres@txAngleF                = -90.               ;-- rotate the text

  gsn_text_ndc(wks,"y-axis", 0.84, 0.5, ndcres)      ;-- draw a axis label on the right

;---------------------------------------------------------
;-- draw the plot
;---------------------------------------------------------
  draw(plot)
  frame(wks)
  delete(wks)
```

```
;-------------------------------------------------------------
;-- add a logo to the finished plot (upper left corner)
;-- (this could be done only for PNG plot output)
;-------------------------------------------------------------
  logo = "$HOME/Pictures/DKRZ/DKRZ_Logo_mit_Text.jpg"

  if(itype .eq."pdf" .or. itype.eq."ps") then
     cmd  = "composite -geometry 80x80+20+20 "+logo+" plot_text_tipps."+\
            itype+" plot_text_tipps."+otype
  else
     cmd  = "composite -geometry 300x300+70+130 "+logo+" plot_text_tipps."+\
            itype+" plot_text_tipps."+otype
  end if

  system(cmd)

end
```